
Abstract

Topological groups are blends of groups and topological spaces with the property that
the multiplication and inversion operations are continuous functions. They frequently
occur in mathematics and physics, e.g. in the form of Lie groups. We have developed
a formalization of topological groups in the theorem prover Isabelle. It contains
basic properties of topological groups as well as their uniform structures. The most
notable formalized result is the Birkhoff-Kakutani theorem which gives a necessary
and sufficient condition for the metrizability of a topological group. We also give
examples for topological groups such as Rn, R× and the general linear group GLn(R)

with its subgroups. In this thesis we present our formalization and describe the proof
of Birkhoff-Kakutani in detail.

1 Introduction

Algebraic objects like groups or vector spaces often have a natural topological structure
which is, in some sense, compatible with the algebraic structure. It can be very fruitful
to analyze the interplay of algebraic and topological properties of such objects instead
of focusing solely on one of these two domains. This motivates defining blends of alge-
braic and topological structures satisfying certain compatibility axioms. An example
for such a blend is the concept of topological groups. [1, foreword] A topological group
is a group G with a topology T endowed on G such that both the group multiplication
G× G → G, (σ, τ) 7→ στ and the inverse mapping G → G, σ 7→ σ−1 are continuous
functions with respect to T. Here, the topology on G×G is the product topology. Some
authors additionally require G to be a Hausdorff space. First examples of topological
groups are (Rn,+) and (R×, ·) as well as the general linear groups GLn(R), GLn(C)

and their subgroups. [1, Chapter 1.2] Moreover, all Lie groups are topological groups. Lie
groups are groups which are also differentiable manifolds such that both multiplication
and inversion are smooth (i.e. C∞) functions. [2, page 31] Lie groups, in particular the
matrix Lie groups, appear frequently in physics (especially in particle physics), where
they are used to study symmetries. Matrix Lie groups are the groups GLn(R), GLn(C)

and their closed subgroups. For instance, the special orthogonal group SOn ⊆ GLn(R)

is a matrix Lie group consisting of the rotations of the n-dimensional space Rn around
the origin. It often occurs in physics problems involving rotational symmetry. [3]

An important motivation for studying topological groups are Haar measures, which
are measures on topological groups with properties similar to the Lebesgue measure
on Rn. Haar measures enable us to integrate over topological groups in a way that is,
in some sense, compatible with the group structure. Formally, a left Haar measure is a
regular left-invariant Borel measure µ on a locally compact group G. A locally compact
group is a topological group which is a locally compact Hausdorff space. Regularity of
a measure is a property of compatibility with the underlying topology: For every Borel
set A ⊆ G,

µ(A) = inf
A⊆U, U open

µ(U) = sup
K⊆A, K compact

µ(K).

Left-invariance, on the other hand, is about compatiblity with the group structure:
For every Borel set A ⊆ G and group element σ ∈ G, we have µ(σA) = µ(A) where

1 Introduction

σA := {στ | τ ∈ A}. The notions of right-invariance and right Haar measure are
defined analogously. It can be shown that Haar measures µ satisfy µ(U) > 0 for
non-empty open sets U ⊆ G and µ(K) < ∞ for compact sets K ⊆ G. The left-invariance
of left Haar measures µ implies the following property of integrals with respect to µ:
Let f : G → R be a function and τ ∈ G a group element. Then f is µ-integrable if and
only if G → R, σ 7→ f (τσ) is, and in this case∫

G
f (τσ) dµ(σ) =

∫
G

f (σ) dµ(σ).

It can be shown that left and right Haar measures exist on all locally compact groups
and both measures are unique up to constant multiples. For example, the Lebesgue
measure λn is a left and right Haar measure on Rn. Furthermore,

µ(A) :=
∫

A
|det(X)|−n dλn2

(X)

for Borel sets A ⊆ GLn(R) defines a left and right Haar measure on GLn(R). Here, λn2

is the Lebesgue measure on the matrix space Rn×n which can be identified with Rn2
.

[2, pages 3-5]

The above motivation shows that topological groups are an interesting and important
topic in mathematics. We have formalized the basic theory of topological groups in
the theorem prover Isabelle [4]. Our formalization [5] is based on the HOL-Algebra
and HOL-Analysis libraries. It proves basic properties of topological groups and gives
examples for such groups. It also includes the left and right uniformities of topological
groups. For this purpose, we have given a definition of uniform structures which is
not based on type classes, in contrast to the definition from HOL-Analysis. The most
important theorem proven is the Birkhoff-Kakutani theorem which gives a necessary and
sufficient condition for the metrizability of topological groups. In the following, we first
describe our formalization and then explain the proof of Birkhoff-Kakutani in detail.

2 Description of formalization

In this chapter we give an overview of our formalization and present the most important
definitions and results. The section proving the Birkhoff-Kakutanti theorem is excluded
and discussed in Chapter 3 instead. Table 2.1 provides information on relevant Isabelle
notation from HOL-Algebra and HOL-Analysis.

2.1 Definition and basic results

There are two implementations of groups in HOL-Algebra: a type class and a lo-
cale. Likewise, there are two notions of topologies in HOL-Analysis: a type class
topological_space and a type ’a topology. The weakness of the type class defini-
tions is that one cannot define groups or topologies on proper subsets of a type’s
universe, resulting in the lack of fundamental concepts like (normal) subgroups. There-
fore we avoid the type classes. We give the following Isabelle definition of topological
groups:

locale topological-group = group +
fixes T :: ′g topology
assumes group-is-space: topspace T = carrier G
assumes inv-continuous: continuous-map T T (λσ. inv σ)
assumes mul-continuous: continuous-map (prod-topology T T) T (λ(σ,τ). σ⊗τ)

A trivial consequence of the group_is_space assumption is that open sets are subsets
of the group’s carrier set. This fact will be used throughout the formalization to connect
topological and algebraic arguments.

lemma open-set-in-carrier:
assumes openin T U
shows U ⊆ carrier G

using openin-subset group-is-space assms by auto

It follows immediately from the continuity of the multiplication operation that left
translations G → G, τ 7→ στ as well as right translations G → G, τ 7→ τσ by some
group element σ ∈ G are continuous. In fact, translations are even homeomorphisms
since their inverse functions are translations again. The inverse mapping is also a

2 Description of formalization

homeomorphism because it is its own inverse function. [6, Lemma 1.4] Moreover,
conjugations G → G, τ 7→ στσ−1 by some element σ ∈ G are homeomorphisms since
they are compositions of translations.

lemma translations-continuous:
assumes in-group: σ ∈ carrier G
shows continuous-map T T (λτ. σ⊗τ) and continuous-map T T (λτ. τ⊗σ)

lemma translations-homeos:
assumes in-group: σ ∈ carrier G
shows homeomorphic-map T T (λτ. σ⊗τ) and homeomorphic-map T T (λτ. τ⊗σ)

abbreviation conjugation :: ′g⇒ ′g⇒ ′g where
conjugation σ τ ≡ σ ⊗ τ ⊗ inv σ

corollary conjugation-homeo:
assumes in-group: σ ∈ carrier G
shows homeomorphic-map T T (conjugation σ)

lemma inverse-homeo: homeomorphic-map T T (λσ. inv σ)

Furthermore, translations being homeomorphisms implies the translation invariance
of the topology: The image σU = { στ | τ ∈ U } of an open subset U ⊆ G under left
translation by an element σ ∈ G is open again. This is still true if we exchange left with
right and/or open with closed. [6, Corollary 1.5]

corollary open-set-translations:
assumes open-set: openin T U and in-group: σ ∈ carrier G
shows openin T (σ <# U) and openin T (U #> σ)

corollary closed-set-translations:
assumes closed-set: closedin T U and in-group: σ ∈ carrier G
shows closedin T (σ <# U) and closedin T (U #> σ)

2 Description of formalization

Table 2.1: List of relevant operators from HOL-Algebra and HOL-Analysis

Operator Meaning

carrier G carrier set of a group
topspace T carrier set of a topology
x ⊗ y group multiplication operation
inv x group inversion operation
1 neutral element of group
N ◁ G normal subgroup
x <# S multiplication of element with set
S #> x multiplication of set with element
S <#> T multiplication of set with set
openin T U true if U is open set in topology T
closedin T U true if U is closed set in topology T
T closure_of S closure of set S in topology T
continuous_map T1 T2 f true if f is a continuous map T1 → T2

connected_component_of_set T x connected component of x in topology T
prod_topology T1 T2 product of two topologies
product_topology T I product of the topologies T(i) with i ∈ I
euclidean topology induced by topological_space

type class
’a^’n vector type of dimension n over ’a
v $ i i-th component of vector v
’a^’n^’m type of matrices m rows, n columns

and entries of type ’a
A $ i $ j entry in i-th row and j-th column

of matrix A
A ** B matrix product
matrix_inv A inverse matrix
mat 1 identity matrix
bdd-above S true if set S is bounded from above
bdd-below S true if set S is bounded from below
Metric-space.mtopology X d metric topology of metric d on X

2 Description of formalization

2.2 Subspaces and quotient spaces

Keeping in mind the homeomorphisms identified above and making use of the fact that
connected components are mapped to connected components under homeomorphisms,
it is not hard to obtain the following result: The connected component of the neutral
element 1 ∈ G is closed under multiplication, inverses and conjugation and is thus a
normal subgroup of G. [1, Proposition 1.4.26] Furthermore, it is a closed subset of G
since connected components are closed in general.

lemma connected-components-homeo:
assumes homeo: homeomorphic-map T1 T2 φ and in-space: x ∈ topspace T1
shows φ‘(connected-component-of-set T1 x) = connected-component-of-set T2 (φ x)

abbreviation connected-component-1 :: ′g set where
connected-component-1 ≡ connected-component-of-set T 1

lemma connected-component-1-props:
shows connected-component-1 ◁ G and closedin T connected-component-1

Let Zσ be the connected component of σ ∈ G. The coset Z1σ is the image of Z1 under
right translation by σ and thus a connected component by connected-components-homeo.
Moreover, σ = 1 · σ ∈ Z1σ and thus Z1σ = Zσ. Therefore the quotient group G/Z1

consists of the connected components of G. For example, consider the topological
group R× = R \ {0} with the usual multiplication and the subtopologoy induced by
the euclidean topology on R. The neutral element is the real number 1 ∈ R and its
connected component is the set of positive real numbers R+. The quotient group is
R×/R+ = {R+, R−} ∼= {1,−1}.

Next, we show that subgroups and quotient groups of topological groups are again
topological groups with the respective induced topologies. For subgroups, this is
straightforward [1, Chapter 1.2].

lemma topological-subgroup:
assumes subgroup H G
shows topological-group (G (|carrier := H|)) (subtopology T H)

For quotient groups, we first need quotient topologies. To our knowledge, only the
notion of quotient map has been introduced in HOL-Analysis, but not the notion of
quotient topology. Let X be a topological space and q : X → Y a surjection. According
to the usual definition, a subset U ⊆ Y is open in the quotient topology of X under q
if and only if the preimage q−1(U) is open in X. Maps with this property are called
quotient maps. Our Isabelle defintion looks as follows:

2 Description of formalization

definition quot-topology :: ′a topology⇒ ( ′a⇒ ′b)⇒ ′b topology where
quot-topology T q = topology (λU. U ⊆ q‘(topspace T) ∧ openin T {x ∈ topspace T. q x ∈ U})

It remains to be shown that the defined object actually fulfills the desired properties.
This mainly requires proving that the given open set predicate satisfies the axioms of a
topology. Then, we obtain the following properties of quotient topologies:

lemma quot-topology-open:
fixes T :: ′a topology and q :: ′a⇒ ′b
defines openin-quot U ≡ U ⊆ q‘(topspace T) ∧ openin T {x ∈ topspace T. q x ∈ U}
shows openin (quot-topology T q) = openin-quot

lemma projection-quotient-map: quotient-map T (quot-topology T q) q

corollary topspace-quot-topology: topspace (quot-topology T q) = q‘(topspace T)

In HOL-Algebra, the quotient group of a normal subgroup H ⊴ G has been defined
as the set of right cosets of H. It should be endowed with the quotient topology induced
by the projection map π : G → G/H, σ 7→ Hσ [1, Chapter 1.5]. We define this coset
topology as follows.

abbreviation coset-topology :: ′g set⇒ ′g set topology where
coset-topology H ≡ quot-topology T (r-coset G H)

Now we are ready to prove that quotient groups of topological groups are again
topological groups. To do this, we need the fact that projections of open sets in G and
G× G are open in G/H and G/H × G/H, respectively. Therefore we show that the
projection π : G → G/H is an open map, which is not true for projections in general.
This also implies the openness of G× G → G/H× G/H, (σ, τ) 7→ (π(σ), π(τ)), which
we prove as a general lemma. [7]

lemma projection-open-map:
assumes subgroup: subgroup H G
shows open-map T (coset-topology H) (r-coset G H)

lemma open-map-prod-top:
assumes open-map T1 T3 f and open-map T2 T4 g
shows open-map (prod-topology T1 T2) (prod-topology T3 T4) (λ(x, y). (f x, g y))

lemma topological-quotient-group:
assumes normal-subgroup: N ◁ G
shows topological-group (G Mod N) (coset-topology N)

2 Description of formalization

Often one is only interested in Hausdorff spaces. Next we answer the question when
quotient groups of topological groups are Hausdorff. This is the case if and only if
the corresponding normal subgroup is closed. The implication from right to left is
non-trivial and requires an important auxiliary result: For every neighborhood U ⊆ G
of 1 there is a symmetric neighborhood S of 1 with S2 = {στ | σ, τ ∈ S} ⊆ U [6,
Lemma 1.7]. In this thesis, neighborhoods in topological spaces are always considered
open, so a neighborhood of σ ∈ G is an open set U ⊆ G with σ ∈ U. A subset S ⊆ G is
said to be symmetric if S−1 = {σ−1 | σ ∈ S} ⊆ S. This implies the other inclusion, so
symmetric subsets satisfy S−1 = S.

abbreviation neighborhood :: ′g⇒ ′g set⇒ bool where
neighborhood σ U ≡ openin T U ∧ σ ∈ U

abbreviation symmetric :: ′g set⇒ bool where
symmetric S ≡ {inv σ | σ. σ ∈ S} ⊆ S

lemma neighborhoods-of-1:
assumes neighborhood 1 U
shows ∃V. neighborhood 1 V ∧ symmetric V ∧ V <#> V ⊆ U

Now we can show that the coset space of a closed subgroup is Hausdorff [6, Lemma
1.11].

lemma Hausdorff-coset-space:
assumes subgroup: subgroup H G and H-closed: closedin T H
shows Hausdorff-space (coset-topology H)

The reverse implication is trivial: If G/H is Hausdorff, then H = π−1({H}) is closed
in G since quotient maps are continuous and all points of a Hausdorff space are closed.

lemma Hausdorff-coset-space-converse:
assumes subgroup: subgroup H G
assumes Hausdorff : Hausdorff-space (coset-topology H)
shows closedin T H

corollary Hausdorff-coset-space-iff :
assumes subgroup: subgroup H G
shows Hausdorff-space (coset-topology H)←→ closedin T H

We can also deduce from this that a topological group G is Hausdorff if and only if
1 ∈ G is a closed point since G is trivially homeomorphic to G/{1}.

corollary topological-group-hausdorff-iff-one-closed:
shows Hausdorff-space T ←→ closedin T {1}

2 Description of formalization

Having motivated the importance of closed (normal) subgroups, we continue with
the following result: The closure of a subgroup of a topological group G is again
a subgroup [6, Lemma 1.7]. This immediately implies that the closure of a normal
subgroup is again a normal subgroup. Indeed, if N ⊴ G is normal and φσ : G → G
denotes conjugation by σ ∈ G, then φσ(N) ⊆ φσ(N) ⊆ N since φσ is continuous and
φσ(N) ⊆ N.

lemma subgroup-closure:
assumes H-subgroup: subgroup H G
shows subgroup (T closure-of H) G

lemma normal-subgroup-closure:
assumes normal-subgroup: N ◁ G
shows (T closure-of N) ◁ G

2.3 Uniform structures

2.3.1 General theory

The definitions and results in this section are based on the book Topological and Uniform
Spaces [8, Chapters 7 and 8].

Just as topological spaces generalize metric spaces with respect to continuity, uniform
spaces generalize metric spaces with respect to uniform continuity. In a topological
space X, a neighborhood U of a point x ∈ X is an expression of closeness: points in U
are considered close to x with respect to U. Instead of open sets, a uniform space X
has a set of entourages. An entourage E ⊆ X× X is a relation on X, and points x′ ∈ X
are considered close to x ∈ X with respect to E if xEx′ holds. Consequently, the set of
points close to x with respect to E (the E-neighborhood of x) is E[x] := {x′ ∈ X | xEx′}.
Note that entourages are not necessarily symmetric relations.

The difference between continuity and uniform continuity is a subtle change in the
ordering of quantifiers. A function f : X → Y is continuous if for every x ∈ X and
every neighborhood V of f (x) ∈ Y there is a neighborhood U of x such that f (U) ⊆ V,
i.e. points close to x with respect to U are mapped to points close to f (x) with respect
to V. In contrast, f is uniformly continuous if for every entourage F in Y there is an
entourage E in X such that for every point x ∈ X we have f (E[x]) ⊆ F[ f (x)], i.e. points
close to x with respect to E are mapped to points close to f (x) with respect to F. For
an entourage E, the E-neighborhoods E[x] of all points x should be considered "the

9



2 Description of formalization

same size". Then f being uniformly continuous means that for equally sized neigh-
borhoods F[ f (x1)], F[ f (x2)] there are equally sized neighborhoods E[x1], E[x2] such
that f (E[x1]) ⊆ F[ f (x1)] and f (E[x2]) ⊆ F[ f (x2)]. This motivates the term "uniform
continuity".

A set becomes a uniform space via a uniform structure, which is defined as follows.

Definition 1. A uniform structure consists of a set X, the uniform space, and a predicate
Φ determining which relations E ⊆ X× X are entourages. The entourages defined by
Φ must additionally satisfy the following axioms:

1. There is an entourage.

2. For every entourage E and x ∈ X, (x, x) ∈ E.

3. For every entourage E, E−1 = {(x2, x1) | (x1, x2) ∈ E} is an entourage.

4. For every entourage E, there is an entourage F with

F2 = {(x1, x3) | (x1, x2), (x2, x3) ∈ F} ⊆ E.

5. For every entourage E and relation F ⊆ X× X with E ⊆ F, F is an entourage.

6. For entourages E, F, the intersection E ∩ F is an entourage.

In HOL-Analysis there is a type class uniform_space, but for the reasons de-
scribed earlier this is not suitable for our purposes. Therefore we define a new
type ’a uniformity in analogy to ’a topology from HOL-Analysis. Our formalization
includes an equivalence proof of ’a uniformity and the uniform_space type class.

definition uniformity-on :: ′a set⇒ (( ′a × ′a) set⇒ bool)⇒ bool where
uniformity-on X Φ←→
(∃E. Φ E) ∧
(∀E. Φ E −→ E ⊆ X × X ∧ Id-on X ⊆ E ∧ Φ (E−1) ∧ (∃ F. Φ F ∧ F O F ⊆ E) ∧ (∀ F. E ⊆ F ∧ F
⊆ X × X −→ Φ F)) ∧
(∀E F. Φ E −→ Φ F −→ Φ (E ∩ F))

typedef ′a uniformity = {(X :: ′a set, Φ). uniformity-on X Φ}
morphisms uniformity-rep uniformity

definition uspace :: ′a uniformity⇒ ′a set where
uspace Φ = (let (X, Φ) = uniformity-rep Φ in X)

definition entourage-in :: ′a uniformity⇒ ( ′a × ′a) set⇒ bool where
entourage-in Φ = (let (X, Φ) = uniformity-rep Φ in Φ)

2 Description of formalization

lemma uniformity-inverse ′:
assumes uniformity-on X Φ
shows uspace (uniformity (X, Φ)) = X ∧ entourage-in (uniformity (X, Φ)) = Φ

This is followed by lemmas proving that the entourages of an ’a uniformity object
fulfill the axioms listed above. We then go on to define uniform continuity. In analogy
to regular continuity, uniform continuity can be formulated concisely in the following
way: A function f : X → Y between uniform spaces is uniformly continuous if and
only if for every entourage E of Y, f−1(E) := {(x1, x2) ∈ X× X | ( f (x1), f (x2)) ∈ E} is
an entourage in X.

definition ucontinuous :: ′a uniformity⇒ ′b uniformity⇒ ( ′a⇒ ′b)⇒ bool where
ucontinuous Φ Ψ f ←→

f ∈ uspace Φ→ uspace Ψ ∧
(∀E. entourage-in Ψ E −→ entourage-in Φ {(x, y) ∈ uspace Φ × uspace Φ. (f x, f y) ∈ E})

The uniform structure of a uniform space X induces a topology on X: A subset
U ⊆ X is called open if and only if for every x ∈ U there is an entourage E such
that E[x] ⊆ U. This definition is reminiscent of the topology on metric spaces which
uniform spaces are a generalization of. Topologies induced by uniform structures are
called uniformizable and share some of the nice properties of metric topologies. We
give the following Isabelle definition of the uniform topology.

definition utopology :: ′a uniformity⇒ ′a topology where
utopology Φ = topology (λU. U ⊆ uspace Φ ∧ (∀ x∈U. ∃E. entourage-in Φ E ∧ E‘‘{x} ⊆ U))

As with the definition of quotient topologies we need to prove that our definition
satisfies the axioms of a toplogy. From this we can derive the following desired
properties:

lemma openin-utopology:
fixes Φ :: ′a uniformity
defines uopen U ≡ U ⊆ uspace Φ ∧ (∀ x∈U. ∃E. entourage-in Φ E ∧ E‘‘{x} ⊆ U)
shows openin (utopology Φ) = uopen

lemma topspace-utopology:
shows topspace (utopology Φ) = uspace Φ

It is straightforward to show that uniform continuity implies continuity in the
uniform topologies.

lemma ucontinuous-imp-continuous:
assumes ucontinuous Φ Ψ f
shows continuous-map (utopology Φ) (utopology Ψ) f

11



2 Description of formalization

To conclude this section, we give a first example of a uniform space: Every metric
space (M, d) can be made into a uniform space by defining entourages to be the
supersets of the sets {(x, y) ∈ M×M | d(x, y) < ε} for ε ∈ R+. Uniform continuity of
functions between metric spaces with respect to the uniform structure coincides with
the usual notion of uniform continuity. In the context of the Metric_space locale we
define the following.

abbreviation mentourage :: real⇒ ( ′a × ′a) set where
mentourage ε ≡ {(x,y) ∈ M × M. d x y < ε}

definition muniformity :: ′a uniformity where
muniformity = uniformity (M, λE. E ⊆ M × M ∧ (∃ ε > 0. mentourage ε ⊆ E))

Once again we show the conformity of our definition with the uniformity axioms to
be able to make use of the defined properties. Then it is straightforward to show that
the induced uniform topology is equal to the metric topology.

lemma
uspace-muniformity: uspace muniformity = M and
entourage-muniformity: entourage-in muniformity = (λE. E ⊆ M × M ∧ (∃ ε > 0. mentourage ε ⊆

E))

lemma uniformity-induces-mtopology: utopology muniformity = mtopology

2.3.2 Uniformities on topological groups

The definitions and results in this section are taken from the book Topological Groups
and Related Structures [1, Chapter 1.8].

In a topological group G, the elements of some neighborhood U of the neutral
element 1 ∈ G can be considered "close" to 1 or "small" with respect to U. It is thus
intuitive to consider elements τ ∈ G "close" to σ ∈ G with respect to U if σ−1τ ∈ U.
This leads to the definition of a uniform structure on G where entourages are the
supersets of the sets {(σ, τ) ∈ G× G | σ−1τ ∈ U} with U a neighborhood of 1. This
uniformity is called left uniformity to distinguish it from the right uniformity which
regards τ ∈ G as "close" to σ ∈ G if στ−1 ∈ U.

abbreviation left-entourage :: ′g set⇒ ( ′g × ′g) set where
left-entourage U ≡ {(σ,τ) ∈ carrier G × carrier G. inv σ ⊗ τ ∈ U}

definition left-uniformity :: ′g uniformity where left-uniformity =
uniformity (carrier G, λE. E ⊆ carrier G × carrier G ∧ (∃U. neighborhood 1 U ∧ left-entourage U ⊆

E))

12



2 Description of formalization

lemma
uspace-left-uniformity: uspace left-uniformity = carrier G and
entourage-left-uniformity: entourage-in left-uniformity =
(λE. E ⊆ carrier G × carrier G ∧ (∃U. neighborhood 1 U ∧ left-entourage U ⊆ E))

It is not hard to conclude that the induced uniform topology is equal to the original
topology on the group. Topological groups are thus uniformizable spaces.

lemma left-uniformity-induces-group-topology:
shows utopology left-uniformity = T

Analogous definitions and lemmas are given for the right uniformity. We conclude
this section with the remark that left translations are uniformly continuous with respect
to the left uniformity, whereas right translations are uniformly continuous with respect
to the the right uniformity.

lemma translations-ucontinuous:
assumes in-group: σ ∈ carrier G
shows ucontinuous left-uniformity left-uniformity (λτ. σ ⊗ τ) and

ucontinuous right-uniformity right-uniformity (λτ. τ ⊗ σ)

2.4 Examples

Any group becomes a topological group by endowing it with the discrete topology
since any function defined on a discrete space is continuous [1, Chapter 1.2]. This is
the most trivial example of topological groups.

lemma (in group) discrete-topological-group:
shows topological-group G (discrete-topology (carrier G))

A more interesting example is the additive group of the vector space Rn. Its topology
is the product topology induced by the usual topology on R. [1, Chapter 1.2] The
continuity of addition and negation, the inverse mapping, is well-known. HOL-Analysis
implements cartesian vector spaces by means of a type ’a^’n, where ’a is an arbitrary
base type and ’n is a finite type. For example, the type real^3 models R3 using
the three-element-type 3. If the type ’a belongs to the topological_space type class
like real does, then ’a^’n also implements topological_space and its topology is the
product topology. If a type ’a implements topological_space, there is an ’a topology
object called euclidean connecting the type class to the set-based notion of topologies.
We use this topology for our formalization. The continuity of addition and negation
has been proven in HOL-Analysis.

2 Description of formalization

lemma topological-group-real-power-space:
defines R :: (real^ ′n) monoid ≡ (|carrier = UNIV, monoid.mult = (+), one = 0|)
defines T :: (real^ ′n) topology ≡ euclidean
shows topological-group R T

The unit group R× = R \ {0} of the reals with the usual multiplication and the
subtopology induced by the standard topology on the reals also forms a topological
group [1, Chapter 1.2].

definition unit-group :: ( ′a :: field) monoid where
unit-group = (|carrier = UNIV − {0}, mult = (∗), one = 1|)

lemma
group-unit-group: group unit-group and
inv-unit-group: x ∈ carrier unit-group =⇒ invunit-group x = inverse x

lemma topological-group-real-unit-group:
defines T :: real topology ≡ subtopology euclidean (UNIV − {0})
shows topological-group unit-group T

Our most important examples of topological groups are the general linear group
GLn(R) = {A ∈ Rn×n | det(A) ̸= 0} and its subgroups. One can give the matrix space
Rm×n a natural topological structure by interpreting it as the vector space Rmn (which,
again, is endowed with the product topology). [1, Chapter 1.2]

2.4.1 Topologies on vector types

HOL-Analysis considers matrices as vectors of vectors. Matrices with components
of type ’a are modelled by the type ’a^’n^’m where m is the number of rows and
n is the number of columns. In order to prove the continuity of functions between
vector types, we need to exploit the fact that the euclidean topology on ’a^’n is a
product topology of the euclidean topology on ’a. However, euclidean on ’a^’n is
not actually defined via the product_topology operator, so we cannot apply any of
the results proven about product_topology. We address this problem by defining a
function vec_topology which takes an ’a topology and builds an (’a^’n) topology
via the product_topology. The product_topology is an (’n => ’a) topology, not an
(’a^’n) topology, but ’a^’n is just a typedef of ’n => ’a. We use the morphism
vec_lambda from ’n => ’a to ’a^’n and our quot_topology operator to induce an
(’a^’n) topology from the product_topology.

definition vec-topology :: ′a topology⇒ ( ′a^ ′n) topology where
vec-topology T = quot-topology (product-topology (λi. T) UNIV) vec-lambda

2 Description of formalization

The morphism vec_lambda is a bijective quotient map from the product_topology
to the vec_topology and thus a homeomorphism, along with the inverse morphsim
vec_nth.

lemma producttop-vectop-homeo:
shows homeomorphic-map (product-topology (λi. T) UNIV) (vec-topology T) vec-lambda

lemma vectop-producttop-homeo:
shows homeomorphic-map (vec-topology T) (product-topology (λi. T) UNIV) vec-nth

We can now formally prove our claim that the euclidean topology on ’a^’n is a
product topology of the euclidean topology on ’a. The proof does not contain any
new ideas, it is just going back and forth between the two implementations of product
topologies.

lemma vec-topology-euclidean:
defines T :: ( ′a :: topological-space) topology ≡ euclidean
defines Tvec :: ( ′a^ ′n) topology ≡ euclidean
shows vec-topology T = Tvec

The product topology on the product space ∏i∈I Xi has the following two essential
properties.

1. All projections πj : ∏i∈I Xi → Xj with j ∈ I are continuous.

2. A function f : Y → ∏i∈I Xi is continuous if for all j ∈ I, πj ◦ f is continuous
(universal property of the product):

∏i∈I Xi

Y Xj

πj
f

πj◦ f

The homeomorphisms from above allow us to transfer these two properties to the
vec_topology.

lemma vec-projection-continuous:
shows continuous-map (vec-topology T) T (λv. v$i)

lemma vec-components-continuous-imp-continuous:
fixes f :: ′x⇒ ′a^ ′n
assumes ∀ i. continuous-map X T (λx. (f x) $ i)
shows continuous-map X (vec-topology T) f

2 Description of formalization

We can now apply these results to the euclidean topology on ’a^’n because of the
lemma vec_topology_euclidean. For convenience we also define a matrix_topology
on the type of matrices and give matrix versions of these lemmas. The proofs just apply
the above results two times.

definition matrix-topology :: ′a topology⇒ ( ′a^ ′n^ ′m) topology where
matrix-topology T = vec-topology (vec-topology T)

lemma matrix-topology-euclidean:
shows matrix-topology euclidean = euclidean

lemma matrix-projection-continuous:
shows continuous-map (matrix-topology T) T (λA. A$i$j)

lemma matrix-components-continuous-imp-continuous:
fixes f :: ′x⇒ ′a^ ′n^ ′m
assumes

∧
i j. continuous-map X T (λx. (f x) $ i $ j)

shows continuous-map X (matrix-topology T) f

Another advantage of the vec_topology operator is that it allows endowing ’a^’n
with product topologies induced by topologies on ’a other than euclidean.

2.4.2 Matrix groups

HOL-Analysis proves that sums, products, etc. of continuous functions into the reals
are continuous. Using this and the above lemmas, it is not hard to show that the
determinant function as well as matrix transposition and multiplication are continuous.

lemma det-continuous:
defines T :: (real^ ′n^ ′n) topology ≡ euclidean
shows continuous-map T euclideanreal det

lemma transpose-continuous:
shows continuous-map (euclidean :: (( ′a :: topological-space)^ ′n^ ′m) topology) euclidean transpose

lemma matrix-mul-continuous:
defines T1 :: (real^ ′n^ ′m) topology ≡ euclidean
defines T2 :: (real^ ′r^ ′n) topology ≡ euclidean
defines T3 :: (real^ ′r^ ′m) topology ≡ euclidean
shows continuous-map (prod-topology T1 T2) T3 (λ(A,B). A ∗∗ B)

We define the general linear group.

definition GL :: (( ′a :: field)^ ′n^ ′n) monoid where
GL = (|carrier = {A. invertible A}, monoid.mult = (∗∗), one = mat 1|)

2 Description of formalization

definition GL-topology :: (real^ ′n^ ′n) topology where
GL-topology = subtopology euclidean (carrier GL)

lemma
GL-group: group GL and
GL-carrier [simp]: carrier GL = {A. invertible A} and
GL-inv [simp]: A ∈ carrier GL =⇒ invGL A = matrix-inv A

Showing that GLn(R) is a topological group still requires the continuity of matrix
inversion. The functions discussed above are defined explicitly via formulas, therefore
we can systematically prove them to be continuous. Matrix inversion, however, is
defined non-constructively via the characteristic property AA−1 = A−1A = 1. Showing
its continuity thus requires finding an explicit representation of the inverse matrix. This
can be done with the help of Cramer’s Rule, formalized in HOL-Analysis. We call the
explicit inversion function cramer_inv and prove its correctness.

lemma cramer-inv-is-inverse:
assumes invertible: invertible (A :: ( ′a :: field)^ ′n^ ′n)
shows matrix-inv A = cramer-inv A

lemma matrix-inv-continuous:
shows continuous-map (GL-topology :: (real^ ′n^ ′n) topology) GL-topology matrix-inv

We can now conclude that GLn(R) is actually a topological group. Moreover, it is
open in the space of n× n matrices since it is the preimage of the open set R \ {0}
under the continuous determinant map.

lemma
GL-topological-group: topological-group GL GL-topology and
GL-open: openin (euclidean :: (real^ ′n^ ′n) topology) (carrier GL)

It follows that all subgroups of GLn(R) are topological groups, too. An example is the
special linear group SLn(R) = {A ∈ Rn×n | det(A) = 1}. It is a subgroup because the
determinant det : GLn(R) → R× is a group homomorphism and SLn(R) = ker(det).
Moreover, SLn(R) ⊆ GLn(R) is a closed subset because it is the preimage of the closed
point 1 ∈ R under the continuous determinant map. [6, Example 1.3]

lemma det-homomorphism: group-hom GL unit-group det

definition SL :: (( ′a :: field)^ ′n^ ′n) monoid where
SL = GL (|carrier := {A. det A = 1}|)

2 Description of formalization

lemma
SL-kernel-det: carrier (SL :: (( ′a :: field)^ ′n^ ′n) monoid) = kernel GL unit-group det and
SL-subgroup: subgroup (carrier SL) (GL :: ( ′a^ ′n^ ′n) monoid) and
SL-carrier [simp]: carrier SL = {A. det A = 1}

lemma
SL-topological-group: topological-group SL (subtopology GL-topology (carrier SL)) and
SL-closed: closedin GL-topology (carrier SL)

Analogous results are proven for the subgroup On of orthogonal n× n matrices and
the subgroup SOn = SLn(R) ∩On.

3 The Birkhoff-Kakutani theorem

The most important result in our formalization is the Birkhoff-Kakutani theorem. In
this chapter, we present informal proofs of this theorem and the preliminary results
it depends on. We will also compare the most interesting proof to its formalized
version. The presented results (and their proofs, except for the proof of Lemma 1)
have been taken and adapted from the book Topological Groups and Related Structures [1,
pages 151-155] of Arhangel’skii and Tkachenko. (The authors of this book assume all
topological groups to be Hausdorff spaces.)

Recall that a topological space X is called metrizable if its topology is induced by
a metric on X. Furthermore, X is called first-countable if each point has a countable
neighborhood base, i.e. for every x ∈ X there is a sequence (Un)n∈N of neighborhoods of
x such that for every neighborhood V of x there is an n ∈N with Un ⊆ V.

Definition 2. A metric ∆ : G× G → G on a group is called left-invariant if ∆(ρσ, ρτ) =

∆(σ, τ) for all σ, τ, ρ ∈ G. Likewise, it is right-invariant if ∆(σρ, τρ) = ∆(σ, τ).

Theorem 1 (Birkhoff-Kakutani). A topological group G is metrizable if and only if it is a
first-countable Hausdorff space. In this case, the metric inducing the topology on G can be
chosen to be left-invariant or right-invariant.

Note that left-invariance and right-invariance are notions of compatibility with the
group structure. Every metric space (X, ∆) is a first-countable Hausdorff space: Denote
the open ball around x ∈ X with radius ε ∈ R+ as Bε(x) := {y ∈ X | ∆(x, y) < ε}, then
the sets Un := B1/n(x) form a countable neighborhood base at x. Therefore we only
have to prove that first-countable Hausdorff topological groups admit left-invariant
and right-invariant metrics inducing the group topology. One can obtain a metric ∆ on
a normed vector space by defining ∆(x, y) := ∥x− y∥. The proof we present works in
a similar way. Firstly, it is necessary to generalize the concept of a norm from vector
spaces to groups.

3.1 Prenorms on groups

Definition 3. A function N : G → R on a group G is called a prenorm if it satisfies the
following axioms:

3 The Birkhoff-Kakutani theorem

(i) N(1) = 0 for the neutral element 1 ∈ G
(ii) N(στ) ≤ N(σ) + N(τ) for all σ, τ ∈ G (triangle inequality)
(iii) N(σ−1) = N(σ) for all σ ∈ G

It is clear that for every normed vector space (V, ∥ · ∥), the norm ∥ · ∥ is a prenorm
on the additive group (V,+).

Proposition 1 (Properties of group prenorms). Let N : G → R be a group prenorm. Then
N satisfies the following properties:
Non-negativity: N(σ) ≥ 0 for all σ ∈ G
Reverse triangle inequality: |N(σ)− N(τ)| ≤ N(στ−1) for all σ, τ ∈ G

Proof. We have 0 = N(1) = N(σσ−1) ≤ N(σ) + N(σ−1) = 2N(σ), so N(σ) ≥ 0.
Moreover, N(σ) = N(στ−1τ) ≤ N(στ−1) + N(τ), so N(σ)− N(τ) ≤ N(στ−1). Also,
N(τ) = N(τ−1) ≤ N(σ−1) + N(στ−1), therefore N(τ) − N(σ) ≤ N(στ−1). Thus
|N(σ)− N(τ)| ≤ N(στ−1).

This can be more or less directly translated to Isabelle.

definition group-prenorm :: ( ′g⇒ real)⇒ bool where
group-prenorm N ←→

N 1 = 0 ∧
(∀ σ τ. σ ∈ carrier G ∧ τ ∈ carrier G −→ N (σ ⊗ τ) ≤ N σ + N τ) ∧
(∀ σ ∈ carrier G. N (inv σ) = N σ)

lemma group-prenorm-clauses[elim]:
assumes group-prenorm N
shows

N 1 = 0 and∧
σ τ. σ ∈ carrier G =⇒ τ ∈ carrier G =⇒ N (σ ⊗ τ) ≤ N σ + N τ and∧
σ. σ ∈ carrier G =⇒ N (inv σ) = N σ

using assms unfolding group-prenorm-def by auto

proposition group-prenorm-nonnegative:
assumes prenorm: group-prenorm N
shows ∀ σ ∈ carrier G. N σ ≥ 0

proposition group-prenorm-reverse-triangle-ineq:
assumes prenorm: group-prenorm N and in-group: σ ∈ carrier G ∧ τ ∈ carrier G
shows |N σ − N τ| ≤ N (σ ⊗ inv τ)

Next we need a way to construct prenorms on groups.

Lemma 1 (Induced group prenorm). Let f : G → R be a bounded function on a group G.
Then f induces a prenorm N f : G → R, σ 7→ supτ∈G | f (τσ)− f (τ)|.

3 The Birkhoff-Kakutani theorem

Proof. First note that N f is well-defined: f is bounded, so there is some c ∈ R with
| f (τ) ≤ c| for all τ ∈ G. Then | f (τσ) − f (τ)| ≤ | f (τσ)| + | f (τ)| ≤ 2c is bounded
above, so the supremum exists. Now we need to check the prenorm axioms.
(i) N f (1) = supτ∈G | f (τ)− f (τ)| = 0.
(ii) We have | f (ρστ)− f (ρ)| = | f (ρστ)− f (ρσ) + f (ρσ)− f (ρ)| ≤ | f (ρστ)− f (ρσ)|+
| f (ρσ)− f (ρ)| ≤ N f (τ) + N f (σ) for all ρ ∈ G. Thus N f (στ) ≤ N f (σ) + N f (τ) by the
supremum property.
(iii) | f (τσ−1) − f (τ)| = | f (τσ−1σ) − f (τσ−1)| for all τ ∈ G and | f (ρσ) − f (ρ)| =
| f (ρσσ−1)− f (ρσ)| for all ρ ∈ G. Therefore N(σ−1) and N(σ) are supremums over the
same set, meaning N(σ−1) = N(σ).

In the Isabelle formalization we add a lemma stating that the expression within the
supremum is bounded. We need this fact whenever we want to use the property that
the supremum is an upper bound.

definition induced-group-prenorm :: ( ′g⇒ real)⇒ ′g⇒ real where
induced-group-prenorm f σ = (SUP τ ∈ carrier G. |f (τ ⊗ σ) − f τ|)

lemma induced-group-prenorm-welldefined:
fixes f :: ′g⇒ real
assumes f-bounded: ∃ c.∀ τ ∈ carrier G. |f τ| ≤ c and in-group: σ ∈ carrier G
shows bdd-above ((λτ. |f (τ ⊗ σ) − f τ|)‘(carrier G))

lemma bounded-function-induces-group-prenorm:
fixes f :: ′g⇒ real
assumes f-bounded: ∃ c.∀ σ ∈ carrier G. |f σ| ≤ c
shows group-prenorm (induced-group-prenorm f )

The translation invariance of the topology on a topological group G implies the
following sufficient condition for the continuity of a prenorm N : G → R.

Proposition 2. For the continuity of a prenorm N : G → R on a topological group G, it is
already sufficient that N is continuous at 1 ∈ G, i.e. for every ε ∈ R+ there is a neighborhood
U of 1 with N(σ) < ε for all σ ∈ U.

Proof. We are done if we can find a neighborhood V of σ with N(V) ⊆ Bε(N(σ)) for
every σ ∈ G and every ε ∈ R+. Fix some σ ∈ G and ε ∈ R+. From the assumption
we obtain a neighborhood U of 1 with N(τ) < ε for all τ ∈ U. By the translation
invariance of the topology on G, σU is a neighborhood of σ · 1 = σ. Let τ ∈ U. We have
|N(σ)− N(στ)| = |N(σ−1)− N((στ)−1)| ≤ N(σ−1στ) = N(τ) < ε using the reverse
triangle inequality. Thus N(σU) ⊆ Bε(N(σ)).

proposition group-prenorm-continuous-if-continuous-at-1:
assumes prenorm: group-prenorm N and

continuous-at-1: ∀ ε>0.∃U. neighborhood 1 U ∧ (∀ σ∈U. N σ < ε)
shows continuous-map T euclideanreal N

3.2 A prenorm respecting the group topology

As suggested earlier, we want to define a metric ∆ on the topological group G by setting
∆(σ, τ) := N(στ−1) for some prenorm N. ∆ should induce the group topology, but
we can only hope for this if ∆(σ, τ) being small coincides with τ being in a "small"
neighborhood of σ. Therefore, N(ρ) has to be small if and only if ρ is in a "small"
neighborhood of 1. The following technical result constructs such prenorms N.

Lemma 2 (Construction of prenorms respecting the group topology). Let G be a topologi-
cal group and (Un)n∈N a sequence of symmetric neighborhoods of 1 ∈ G with U2

n+1 ⊆ Un for
all n ∈N. Then there is a prenorm N : G → R such that, for all n ∈N,

{σ ∈ G | N(σ) < 1/2n} ⊆ Un ⊆ {σ ∈ G | N(σ) ≤ 2/2n}.

Proof. The first step is to associate every positive dyadic rational number m/2n where
m ∈N+, n ∈N with a neighborhood of 1. This is done by means of a function V(m, n)
with the property V(m, n) = V(m′, n′) if m/2n = m′/2n′ . V is defined by recursion
over n:

V(m, n) :=


Un if m = 1

G else if m > 2n

V(m/2, n− 1) else if m even

V((m− 1)/2, n− 1) ·Un else.

Note: For V1, V2 neighborhoods of 1, V1V2 = {στ | σ ∈ V1, τ ∈ V2} =
⋃

σ∈V1
σV2 is open

and 1 = 1 · 1 ∈ V1V2, so V1V2 is a neighborhood of 1. Using this fact, it is simple to
show inductively that V(m, n) ⊆ G is actually a neighborhood of 1.

Claim 1: V(m, n) ·Un ⊆ V(m + 1, n) for m ∈N+, n ∈N. We prove this by induction
on n. For n = 0, V(m + 1, 0) = G, so the statement is true. Now assume that the
claim holds for some n. We need to show V(m, n + 1) ·Un+1 ⊆ V(m + 1, n + 1). We
can assume m + 1 ≤ 2n+1, otherwise V(m + 1, n + 1) = G and the statement is trivially
true. Moreover, m > 1 without loss of generality because

V(1, n + 1) ·Un+1 = U2
n+1 ⊆ Un = V(1, n) = V(2, n + 1).

We make a case distinction:
(a) m is even. Then V(m, n + 1) ·Un+1 = V(m/2, n) ·Un+1 = V(m + 1, n + 1).
(b) m is odd. Then

V(m, n + 1) ·Un+1 = V((m− 1)/2, n) ·Un+1 ·Un+1 = V((m− 1)/2, n) ·U2
n+1

⊆ V((m− 1)/2, n) ·Un ⊆ V((m− 1)/2 + 1, n)

= V((m + 1)/2, n) = V(m + 1, n + 1)

where V((m− 1)/2, n) ·Un ⊆ V((m− 1)/2 + 1, n) by the induction hypothesis.

Claim 2: For positive dyadic rationals m1/2n1 ≤ m2/2n2 , we have V(m1, n1) ⊆
V(m2, n2), i.e. V is monotone. First note that V(m, n) = V(m, n) · 1 ⊆ V(m, n) ·Un ⊆
V(m + 1, n) by Claim 1. By induction we obtain V(m, n) ⊆ V(m + k, n) for k ∈ N, so
V(m, n) ⊆ V(m′, n) if m ≤ m′. By induction it is also clear that V(m, n) = V(m · 2k, n +

k) for k ∈N. We make a case distinction.
(a) n1 ≤ n2. Then m1/2n1 = (m1 · 2n2−n1)/2n2 ≤ m2/2n2 , thus m1 · 2n2−n1 ≤ m2, meaning

V(m1, n1) = V(m1 · 2n2−n1 , n1 + n2 − n1) = V(m1 · 2n2−n1 , n2) ⊆ V(m2, n2)

(b) n2 ≤ n1. Proceed analogously to (a).

Now we define f : G → R, σ 7→ inf{m/2n | σ ∈ V(m, n)}. The infimum exists since
the set is bounded below by 0 and also non-empty, as σ ∈ G = V(2, 0). Clearly f (σ) ≥ 0
and f (σ) ≤ 2/20 = 2. So f is bounded and thus induces a prenorm N f : G → R.

Claim 3: For σ ∈ G with f (σ) < m/2n, we have σ ∈ V(m, n). Indeed, by the infimum
property, there must be m′, n′ with m′/2n′ < m/2n and σ ∈ V(m′, n′). Claim 2 yields
V(m′, n′) ⊆ V(m, n), so σ ∈ V(m, n).

Note that f (1) = 0. Otherwise f (1) > 0, so there is some n ∈ N with 1/2n <

f (1). But 1 ∈ Un = V(1, n), so f (1) ≤ 1/2n - a contradiction. Now we prove
{σ ∈ G | N f (σ) < 1/2n} ⊆ Un for n ∈ N. Fix σ ∈ G with N f (σ) < 1/2n. We have
f (σ) = | f (1 · σ)− f (1)| ≤ N f (σ) < 1/2n, thus σ ∈ V(1, n) = Un by Claim 3. It remains
to show the second inclusion.

Let n ∈ N and σ ∈ Un. Additionally, fix some τ ∈ G. Clearly there is k ∈ N+

such that (k − 1)/2n ≤ f (τ) < k/2n. Again, Claim 3 yields τ ∈ V(k, n). We have
σ−1 ∈ Un because σ ∈ Un and Un is symmetric. Therefore τσ, τσ−1 ∈ V(k, n) ·Un ⊆
V(k + 1, n) using Claim 1, and this in turn implies f (τσ), f (τσ−1) ≤ (k + 1)/2n.
Together with (k − 1)/2n ≤ f (τ) from above, this shows (a) f (τσ) − f (τ) ≤ 2/2n

and (b) f (τσ−1)− f (τ) ≤ 2/2n. Now fix some ρ ∈ G. τ represents an arbitrary element
of G, so we can substitute it with ρ in (a) and ρσ in (b). This yields f (ρσ)− f (ρ) ≤ 2/2n

and f (ρ) − f (ρσ) = f (ρσσ−1) − f (ρσ) ≤ 2/2n, implying | f (ρσ) − f (ρ)| ≤ 2/2n. It
follows that N f (σ) ≤ 2/2n since ρ has been arbitrarily chosen. This proves the second
inclusion.

Some intuition for this proof: Claim 2 justifies viewing m/2n as the "size" of V(m, n).
Then f (σ), the infimum of the sizes of neighborhoods of 1 containing σ, can be inter-
preted as the "distance" of σ from 1. Thus f (1) = 0 is intuitively clear. Recall that
N f (σ) = supρ∈G | f (ρσ)− f (ρ)| should be small if and only if σ is in a small neighbor-
hood of 1. This means that for σ in a small neighborhood of 1, i.e. σ ∈ Un for some
large n, f (ρσ) must be close to f (ρ) for any ρ ∈ G. This is achieved by the property
V(m, n) ·Un ⊆ V(m + 1, n) from Claim 1: For every V(m, n) with ρ ∈ V(m, n), this
property implies that ρσ is in V(m + 1, n), and the size difference between V(m, n)
and V(m + 1, n) is 1/2n which is small for large n. The property from Claim 1 is thus
essential for the proof and motivates the definition of V.

In our formalization we create a new context for Lemma 2.

context
fixes U :: nat⇒ ′g set
assumes U-neighborhood: ∀ n. neighborhood 1 (U n)
assumes U-props: ∀ n. symmetric (U n) ∧ (U (n + 1)) <#> (U (n + 1)) ⊆ (U n)

We use the fun command to define the recursive function V. We set V(0, n) := ∅
since Isabelle functions must be total and this is the most convenient value.

private fun V :: nat⇒ nat⇒ ′g set where
V m n = (

if m = 0 then {} else
if m = 1 then U n else
if m > 2^n then carrier G else
if even m then V (m div 2) (n − 1) else
V ((m − 1) div 2) (n − 1) <#> U n

)

The fact that V(m, n) is a neighborhood of 1 is actually not necessary for the proof of
Lemma 2, we only need the fact V(m, n) ⊆ G. This can be trivially shown by induction.

private lemma U-in-group: U k ⊆ carrier G

3 The Birkhoff-Kakutani theorem

private lemma V-in-group:
shows V m n ⊆ carrier G

Claim 1 and Claim 2 are proven as separate lemmas within the context.

private lemma V-mult:
shows m ≥ 1 =⇒ V m n <#> U n ⊆ V (m + 1) n

private lemma V-mono:
assumes smaller: (real m1)/2^n1 ≤ (real m2)/2^n2 and not-zero: m1 ≥ 1 ∧ m2 ≥ 1
shows V m1 n1 ⊆ V m2 n2

We also need a lemma constructing the number k in the proof of Lemma 2.

private lemma approx-number-by-multiples:
assumes hx: x ≥ 0 and hc: c > 0
shows ∃ k :: nat ≥ 1. (real (k−1))/c ≤ x ∧ x < (real k)/c

proof −
let ?k = ⌊x ∗ c⌋ + 1
have ?k ≥ 1 using assms by simp
moreover from this have real (nat ?k) = ?k by auto
moreover have (?k−1)/c ≤ x ∧ x < ?k/c

using assms by (simp add: mult-imp-div-pos-le pos-less-divide-eq)
ultimately show ?thesis

by (smt (verit) nat-diff-distrib nat-le-eq-zle nat-one-as-int of-nat-nat)
qed

The formal version of the proof of Lemma 2 is structured similarly to the informal
version. In Isabelle, using the lower bound property of the infimum requires the fact
that the set in question is actually bounded below (fact bdd-below in the formal proof
below); this corresponds to the well-definedness argument in the informal proof. The
same is true for the supremum (fact N_welldefined), which is why we need the lemma
induced-group-prenorm-welldefined from above.

lemma construction-of-prenorm-respecting-topology:
shows ∃N. group-prenorm N ∧
(∀ n. {σ ∈ carrier G. N σ < 1/2^n} ⊆ U n) ∧
(∀ n. U n ⊆ {σ ∈ carrier G. N σ ≤ 2/2^n})

proof −
define f :: ′g⇒ real where f σ = Inf {(real m)/2^n | m n. σ ∈ V m n} for σ
define N :: ′g⇒ real where N = induced-group-prenorm f

have σ ∈ V 2 0 if σ ∈ carrier G for σ using that by auto

25



3 The Birkhoff-Kakutani theorem

then have contains-2: (real 2)/2^0 ∈ {(real m)/2^n | m n. σ ∈ V m n} if σ ∈ carrier G for σ
using that by blast

then have nonempty: {(real m)/2^n | m n. σ ∈ V m n} ̸= {} if σ ∈ carrier G for σ using that by
fast

have positive: (real m)/2^n ≥ 0 for m n by simp
then have bdd-below: bdd-below {(real m)/2^n | m n. σ ∈ V m n} for σ by fast
have f-bounds: 0 ≤ f σ ∧ f σ ≤ 2 if hσ: σ ∈ carrier G for σ
proof −

from bdd-below have f σ ≤ (real 2)/2^0 unfolding f-def using cInf-lower contains-2[OF hσ] by
meson

moreover have 0 ≤ f σ using cInf-greatest contains-2[OF hσ] unfolding f-def using positive
by (smt (verit, del-insts) Collect-mem-eq empty-Collect-eq mem-Collect-eq)

ultimately show ?thesis by fastforce
qed
then have N-welldefined: bdd-above ((λτ. |f (τ ⊗ σ) − f τ|) ‘ carrier G) if σ ∈ carrier G for σ

using induced-group-prenorm-welldefined that by (metis (full-types) abs-of-nonneg)

have in-V-if-f-smaller: σ ∈ V m n if hσ: σ ∈ carrier G and smaller: f σ < (real m)/2^n for σ m n
proof −

from cInf-lessD obtain q where hq: q ∈ {(real m)/2^n | m n. σ ∈ V m n} ∧ q < (real m)/2^n
using smaller nonempty[OF hσ] unfolding f-def by (metis (mono-tags, lifting))

then obtain m ′ n ′where hm ′n ′: σ ∈ V m ′ n ′ ∧ q = (real m ′)/2^n ′ by fast
moreover have m ′≥ 1
proof (rule ccontr)

assume ¬ m ′≥ 1
then have V m ′ n ′= {} by force
then show False using hm ′n ′ by blast

qed
moreover have m ≥ 1 using f-bounds smaller hσ

by (metis divide-eq-0-iff less-numeral-extra(3) less-one linorder-le-less-linear nle-le of-nat-0
order-less-imp-le)

ultimately have V m ′ n ′⊆ V m n using V-mono hq U-props open-set-in-carrier by simp
then show ?thesis using hm ′n ′ by fast

qed
have f-1-vanishes: f 1 = 0
proof (rule ccontr)

assume f 1 ̸= 0
then have f 1 > 0 using f-bounds by fastforce
then obtain n where hn: f 1 > (real 1)/2^n
by (metis divide-less-eq-1 of-nat-1 one-less-numeral-iff power-one-over real-arch-pow-inv semiring-norm(76)

zero-less-numeral)
have 1 ∈ V 1 n using U-neighborhood by simp
then have (real 1)/2^n ∈ {(real m)/2^n |m n. 1 ∈ V m n} by fast

then show False using hn cInf-lower bdd-below[of 1] unfolding f-def by (smt (verit, ccfv-threshold))

qed
have in-U-if-N-small: σ ∈ U n if in-group: σ ∈ carrier G and N-small: N σ < 1/2^n for σ n
proof −

have f σ = |f (1 ⊗ σ) − f 1| using in-group l-one f-1-vanishes f-bounds by force
moreover have ... ≤ N σ unfolding N-def induced-group-prenorm-def

using cSUP-upper N-welldefined[OF in-group] by (metis (mono-tags, lifting) one-closed)
ultimately have σ ∈ V 1 n using in-V-if-f-smaller[OF in-group] N-small by (smt (verit) of-nat-1)
then show ?thesis by fastforce

qed
have N-bounds: N σ ≤ 2/2^n if hσ: σ ∈ U n for σ n
proof −

have diff-bounded: f (τ ⊗ σ) − f τ ≤ 2/2^n ∧ f (τ ⊗ inv σ) − f τ ≤ 2/2^n if hτ: τ ∈ carrier G
for τ

proof −
obtain k where hk: k ≥ 1 ∧ (real (k−1))/2^n ≤ f τ ∧ f τ < (real k)/2^n

using approx-number-by-multiples[of f τ 2^n] f-bounds[OF hτ] by auto
then have τ ∈ V k n using in-V-if-f-smaller[OF hτ] by blast
moreover have σ ∈ V 1 n ∧ inv σ ∈ V 1 n using hσ U-props by auto
moreover have V k n <#> V 1 n ⊆ V (k + 1) n

using V-mult U-props open-set-in-carrier hk by auto
ultimately have τ ⊗ σ ∈ V (k + 1) n ∧ τ ⊗ inv σ ∈ V (k + 1) n

unfolding set-mult-def by fast
then have a: (real (k + 1))/2^n ∈ {(real m)/2^n | m n. τ ⊗ σ ∈ V m n}
∧ (real (k + 1))/2^n ∈ {(real m)/2^n | m n. τ ⊗ inv σ ∈ V m n} by fast

then have f (τ ⊗ σ) ≤ (real (k + 1))/2^n
unfolding f-def using cInf-lower[of (real (k + 1))/2^n] bdd-below by presburger

moreover from a have f (τ ⊗ inv σ) ≤ (real (k + 1))/2^n
unfolding f-def using cInf-lower[of (real (k + 1))/2^n] bdd-below by presburger

ultimately show ?thesis using hk
by (smt (verit, ccfv-SIG) diff-divide-distrib of-nat-1 of-nat-add of-nat-diff )

qed
have |f (ϱ ⊗ σ) − f ϱ| ≤ 2/2^n if hϱ: ϱ ∈ carrier G for ϱ
proof −

have in-group: σ ∈ carrier G using hσ U-in-group by fast
then have f (ϱ ⊗ σ ⊗ inv σ) − f (ϱ ⊗ σ) ≤ 2/2^n using diff-bounded[of ϱ ⊗ σ] hϱ m-closed

by fast
moreover have ϱ ⊗ σ ⊗ inv σ = ϱ using m-assoc r-inv r-one in-group inv-closed hϱ by presburger
ultimately have f ϱ − f (ϱ ⊗ σ) ≤ 2/2^n by force
moreover have f (ϱ ⊗ σ) − f ϱ ≤ 2/2^n using diff-bounded[OF hϱ] by fast
ultimately show ?thesis by force

qed
then show ?thesis unfolding N-def induced-group-prenorm-def using cSUP-least carrier-not-empty

by meson
qed

then have U n ⊆ {σ ∈ carrier G. N σ ≤ 2/2^n} for n using U-in-group by blast
moreover have group-prenorm N unfolding N-def

using bounded-function-induces-group-prenorm f-bounds by (metis abs-of-nonneg)
ultimately show ?thesis using in-U-if-N-small by blast

qed

3.3 Proof of Birkhoff-Kakutani

We can now finally prove the Birkhoff-Kakutani theorem.

Proof of Birkhoff-Kakutani. As explained earlier, we are done if we can show that the
topology of a first-countable Hausdorff topological group G is induced by left-invariant
and right-invariant metrics. From the first-countability assumption we obtain a count-
able neighborhood base (Wn)n∈N of 1 ∈ G. Recall that for every neighborhood U of 1
there is a symmetric neighborhood V of 1 with V2 ⊆ U. Note that this implies V ⊆ U.
We construct a sequence (Un)n∈N of symmetric neighborhoods of 1 with Un ⊆ Wn and
U2

n+1 ⊆ Un for all n ∈N by recursively choosing symmetric neighborhoods Un+1 of 1
with U2

n+1 ⊆ Un ∩Wn+1. Clearly, Un ⊆ Wn implies that (Un)n is also a neighborhood
base of 1. From Lemma 2 we obtain a prenorm N : G → R such that

{σ ∈ G | N(σ) < 1/2n} ⊆ Un ⊆ {σ ∈ G | N(σ) ≤ 2/2n}

for all n ∈N. By Proposition 2, N is continuous because for any ε ∈ R+ there is some
n ∈ N with 1/2n < ε and thus N(σ) ≤ 2/2n+1 = 1/2n < ε for every σ ∈ Un+1. As
announced earlier, we define a metric ∆ on G by setting ∆(σ, τ) := N(στ−1). We need
to check the metric axioms.
(i) Non-negativity: ∆(στ) ≥ 0 since prenorms are non-negative.
(ii) Symmetry: ∆(σ, τ) = N(στ−1) = N((στ−1)−1) = N(τσ−1) = ∆(τ, σ).
(iii) Positive Definiteness: ∆(σ, τ) = 0 ⇐⇒ σ = τ. Clearly, ∆(σ, σ) = N(σσ−1) =

N(1) = 0. Conversely, assume ∆(σ, τ) = 0. Then N(στ−1) < 1/2n for all n ∈ N, so
στ−1 ∈ Un for all n. By extension, στ−1 is contained in all neighborhoods of 1 since
(Un)n is a neighborhood base. G is a Hausdorff space, therefore

στ−1 ∈
⋂
{V ⊆ G | V neighborhood of 1} = {1}

which implies σ = τ. This is actually the only point where we need the Hausdorffness
assumption.
(iv) Triangle inequality:

∆(σ, ρ) = N(σρ−1) = N(στ−1τρ−1) ≤ N(στ−1) + N(τρ−1) = ∆(σ, τ) + ∆(τ, ρ).

Moreover, the metric ∆ is right invariant because

∆(σρ, τρ) = N(σρ(τρ)−1) = N(σρρ−1τ−1) = N(στ−1) = ∆(σ, τ).

It remains to show that the metric topology defined by ∆ coincides with the original
topology on G. This can be done by proving that the two topologies share a base. A
base of a topology T is a set B of open sets of T such that for every open set V of T
and point x ∈ V there is some B ∈ B with x ∈ B ⊆ V. It is well-known that the open
balls Bε(σ) = {τ ∈ G | ∆(σ, τ) < ε} form a base B := {Bε(σ) | σ ∈ G, ε ∈ R+} of the
metric topology of ∆. Note that Bε(1) = {σ ∈ G | N(σ) < ε}.

Claim 1: Bε(σ) = Bε(1) · σ.
(⊆) If τ ∈ Bε(σ) then ∆(σ, τ) = N(στ−1) < ε, implying N(τσ−1) = N((στ−1)−1) < ε,
thus τ = τσ−1 · σ ∈ Bε(1) · σ.
(⊇) For ρσ ∈ Bε(1) · σ with ρ ∈ Bε(1) we have ∆(σ, ρσ) = N(σ(ρσ)−1) = N(ρ−1) =

N(ρ) < ε, therefore ρσ ∈ Bε(σ).

Claim 2: B is a base of the original group topology.
The sets Bε(σ) = Bε(1) · σ are open in the group topology by the translation invariance
of the topology because Bε(1) = N−1((−∞, ε)) is the preimage of an open set under a
continuous map. Now let V be an open set of the group topology. Let σ ∈ V. Then
Vσ−1 is a neighborhood of 1, so there is some n ∈ N with B1/2n(1) ⊆ Un ⊆ Vσ−1.
Therefore σ ∈ B1/2n(σ) = B1/2n(1) · σ ⊆ Vσ−1σ = V.

We have proved that the two topologies share the base B, thus they are equal.
This means that the group topology is induced by a right-invariant metric. Note
that we could have chosen ∆ to be left-invariant instead of right-invariant by setting
∆(σ, τ) := N(σ−1τ). Mutatis mutandis, the rest of the proof stays the same.

In our formalization we move the construction of the sequence (Un)n into a separate
lemma. The recursive choosing is done by the lemma dependent_nat_choice from the
HOL library.

lemma first-countable-neighborhoods-of-1-sequence:
assumes first-countable T
shows ∃U :: nat⇒ ′g set.
(∀ n. neighborhood 1 (U n) ∧ symmetric (U n) ∧ U (n + 1) <#> U (n + 1) ⊆ U n) ∧
(∀W. neighborhood 1 W −→ (∃ n. U n ⊆ W))

We split the theorem into two Isabelle theorems because the two propositions about
left-invariant and right-invariant metrics require slightly different arguments through-
out the proof. The Metric_space locale is used for statements about metrics. The

decisive fact that topologies sharing the same base are equal is taken from HOL-
Analysis (lemma openin_topology_base_unique).

definition left-invariant-metric ∆←→ Metric-space (carrier G) ∆ ∧
(∀ σ τ ϱ. σ ∈ carrier G ∧ τ ∈ carrier G ∧ ϱ ∈ carrier G −→ ∆ (ϱ ⊗ σ) (ϱ ⊗ τ) = ∆ σ τ)

definition right-invariant-metric ∆←→ Metric-space (carrier G) ∆ ∧
(∀ σ τ ϱ. σ ∈ carrier G ∧ τ ∈ carrier G ∧ ϱ ∈ carrier G −→ ∆ (σ ⊗ ϱ) (τ ⊗ ϱ) = ∆ σ τ)

theorem Birkhoff-Kakutani-left:
assumes Hausdorff : Hausdorff-space T and first-countable: first-countable T
shows ∃∆. left-invariant-metric ∆ ∧ Metric-space.mtopology (carrier G) ∆ = T

theorem Birkhoff-Kakutani-right:
assumes Hausdorff : Hausdorff-space T and first-countable: first-countable T
shows ∃∆. right-invariant-metric ∆ ∧ Metric-space.mtopology (carrier G) ∆ = T

We conclude this section with the corollary that metrizability of a topological group
is equivalent to first-countability together with the Hausdorff property.

corollary Birkhoff-Kakutani-iff :
shows metrizable-space T ←→ Hausdorff-space T ∧ first-countable T

4 Conclusion

In this thesis we presented our formalization and had a detailed look at the proof
of the Birkhoff-Kakutani theorem. What is new in our work is the formalization of
topological groups which - while present in the libraries of other theorem provers - had
been missing in the Isabelle Archive of Formal Proofs (AFP). Furthermore, the AFP
had not yet included the matrix groups GLn(R), SLn(R), On and SOn. Moreover, the
development of a set-based formalization of uniformities in addition to the type-based
notion in the uniform_space type class is a new contribution and complements the
set-based notion of topologies in HOL-Analysis.

A necessary side effect of working with the group locale instead of the group type
class is that most lemmas and propositions need additional assumptions stating that
the elements in question actually belong to the group. This complicates lemma state-
ments and prolongs proofs. It is also unpleasant that the group and its topology are
represented by different Isabelle objects with - at first sight - different carrier sets.

Topological groups are a large topic in mathematics, so there is plenty of work which
could be done in the future. An important task would be developing the theory of
Haar measures on top of the formalization of measures in HOL-Analysis, in particular
proving their existence and uniqueness. For this one first needs to introduce the notion
of locally compact groups. One could also begin formalizing the theory of Lie groups.
This needs the concept of differentiable manifolds which is already included in the
AFP [9].

AFP Archive of Formal Proofs

